OPTIONS LINESIZE = 80;
OPTIONS PAGESIZE = 3200;
OPTIONS NOCENTER NODATE NONUMBER;
filename DOIS 'C:\Users\Michael D. Bradley\Documents\Five Day Delivery\Case Documents\DOIS FY2009.prn';
*************************************************
Daily Delivery Analysis 

Estimates elasticity of hours with respect to 
volume using all delivery days. 
***********************************************;

********************************************
* Read in data     
* “Date” is the date 
* “Day” is the day of the week            
* “Week” is a numerical variable indicating the week
* “DOW” is a numerical variable indicating the day of the week
* “RTS” is the number of routes on which the volume is delivered
* “HRS_OFC” is the Office hours for that day
* “HRS_ST” is the Street hours for that day
* “CLTR” is the volume of cased letters
* “CFLT” is the volume of cased flats
* “DPS” is the volume of DPS letters
* “FSS” is the volume of FSS flats
* “SEQ” is the volume of sequenced mail
* “PCL” is the volume of parcels

********************************************;
 
********************************************;

Data USPS; Infile DOIS ;

Input Date $ DAY $ WEEK DOW RTS HRS_OFC HRS_ST CLTR CFLT DPS FSS SEQ PCL;

********************************************
* Eliminate Satuday June 13, 2009 which reported 32 million parcels;
* Average parcels per day is about 1.6 million;
******************************************** ;

if HRS_ST eq 842206 then delete;


Proc means;

Data USPS; Set USPS; 
VOLUME=CLTR+CFLT+DPS+FSS+SEQ+PCL;
Data USPSDD; Set USPS;
if HRS_OFC lt 10000 then delete;
ROUTES=rts;
VOLRTS=Volume*Routes;
VOl2=Volume*Volume;
RTS2=Routes*Routes;


********************************************
Estimate full quadratic models 
********************************************;
TITLE1 "FULL QUADRATIC MODEL";
TITLE2 "ESTIMATED USING ALL BUSINESS DAYS, FY2009";
PROC REG DATA=USPSDD OUTEST=QUAD_FULL;
MODEL HRS_ST = VOLUME ROUTES VOLRTS VOL2 RTS2;
MODEL HRS_OFC = VOLUME ROUTES VOLRTS VOL2 RTS2;
RUN; QUIT;


*********************************************
Mean center data and estimate translog models
********************************************;
***********
Mean center
**********;
PROC MEANS NOPRINT DATA=USPSDD;
VAR VOLUME ROUTES VOLRTS VOL2 RTS2 HRS_ST HRS_OFC;
OUTPUT OUT = REGMEAN_FULL MEAN = MVOL MRTS MVOLRTS MVOL2 MRTS2 MHRS_ST MHRS_OFC; RUN;

DATA USPS_FULL;
IF _N_=1 THEN SET REGMEAN_FULL (DROP = _TYPE_); SET USPSDD;
VOL	= VOLUME/MVOL;
LVOL 	= LOG(VOL);
LVOL2	= LVOL**2;	
RTS 	= ROUTES/MRTS;
LRTS	= LOG(RTS);
LRTS2	= LRTS**2;
CROSS	= LVOL*LRTS;
LHRS_ST = LOG(HRS_ST);
LHRS_OFC= LOG(HRS_OFC);
RUN; 


****************************************************
Estimate translog model  
***************************************************;
TITLE1 "FULL TRANSLOG MODEL";
TITLE2 "ESTIMATED USING ALL DELIVERY DAYS, FY2009";
PROC REG DATA=USPS_FULL OUTEST=LOG2_FULL;
MODEL LHRS_ST = LVOL LRTS CROSS LVOL2 LRTS2;
MODEL LHRS_OFC = LVOL LRTS CROSS LVOL2 LRTS2;
RUN; QUIT;


***************************************************
Calculate elasticity of hours with respect to volume
***************************************************;
DATA QUAD_FULL1; 
IF _N_=1 THEN SET REGMEAN_FULL; SET QUAD_FULL (DROP =_TYPE_);
IF _DEPVAR_ = "HRS_ST" THEN ELAS_QUAD = (VOLUME + VOLRTS*MRTS + 2*VOL2*MVOL)*(MVOL/MHRS_ST);
IF _DEPVAR_ = "HRS_OFC" THEN ELAS_QUAD = (VOLUME + VOLRTS*MRTS + 2*VOL2*MVOL)*(MVOL/MHRS_OFC);
RUN;

DATA LOG2_FULL1; SET LOG2_FULL (DROP=_TYPE_);
RENAME LVOL=ELAS_FULL_LOG;
IF _DEPVAR_= "LHRS_ST" THEN _DEPVAR_= "HRS_ST";
IF _DEPVAR_= "LHRS_OFC" THEN _DEPVAR_="HRS_OFC";
RUN;

PROC SORT DATA=QUAD_FULL1;
BY _DEPVAR_; RUN;

PROC SORT DATA=LOG2_FULL1;
BY _DEPVAR_; RUN;

DATA FULL; MERGE QUAD_FULL1 LOG2_FULL1;
BY _DEPVAR_; RUN;

DATA PRINT; SET FULL; 
IF _DEPVAR_ = "HRS_ST" THEN _DEPVAR_ = "Street";
IF _DEPVAR_ = "HRS_OFC" THEN _DEPVAR_ = "Office";
LABEL _DEPVAR_ = "Hours Type"
ELAS_QUAD = "Full Quadratic"
ELAS_FULL_LOG = "Full Translog"; RUN;

TITLE1 "Elasticity of Hours with Respect to Volume";
TITLE2 "(Estimated Using Delivery Days, FY2009)";
PROC PRINT DATA=PRINT NOOBS LABEL;
VAR _DEPVAR_ ELAS_QUAD ELAS_FULL_LOG;
RUN;

